Начисление страховых взносов в ПФ с сумм помощи по уходу за ребенком  по достижению им 3-х летнего возраста.

(согласно ч.6 ст.20 и ч.1.ст.19 Закона Украины № 1058-IV от 09/07/03 и Постановление КМУ №1092 от25/08/04).

Реализация этого пункта закона в ИС-ПРО:

1. Создать новый фонд, который рассчитывается для работников, находящихся в отпуске по уходу за ребенком до 3-х лет. 

2. Это фонд можно настроить с параметром «Рассчитывается для всех работников», или разнести индивидуально каждому работнику, находящемуся в отпуске, на закладку Фонды в Лицевых счетах.

3. Расчет фонда производится программным способом.

4. Установить параметр выполнения программы "Для расчетного периода".

5. Настроить Список датированных величин в Настройке-Справочники-Списки величин для ведения изменений по проценту. Пример списка:

	Дата
	Значение

	01/01/2005
	0.0320

	01/04/2005
	0.0323


6. В примере программы внести изменения: ввести реальный код «Списка величин» и код вида оплаты.

Программа фонда: 

DIM TBL(31)
' Начисление страховых взносов в ПФ с сумм помощи по уходу за ребенком  по 
' достижению им 3-х летнего возраста.
' (согласно ч.6 ст.20 и ч.1.ст.19 Закона Украины № 1058-IV от 09/07/03 и 
' Постановление КМУ №1092 от25/08/04).

' Подсчет кол-ва календарных дней в месяце, занятых отпуском по уходу

' за ребенком. Сумма фонда в программе расчитывается пропорционально

' календарным дням.

for i=1 to 31

 TBL[i]=0

next i
PH_INT3=0 ' Флаг контроля наличия в расчетном листе записи с отпуском по уходу
PH_SRD_SM1=0  ' Поле для расчета исходной суммы фонда.

PH_SRD_SM2=0  ' Поле для сбора суммы по отпуску для исходной суммы фонда.

if RL_RP_FIRST( CUR_RP() )>0 then

' размер минимальной зарплаты в расчетном периоде.

   PH_INT2=MONTH_SIZE (MONTH(VO_DAT_RP) , YEAR(VO_DAT_RP) )

   PH_DATE1=VO_DAT_RP

    do

     IF (VOP_MET( VO_CD ) = 43 or VOP_MET( VO_CD ) = 85) AND VO_DAT_RP >= &D01/01/2005 AND VO_PL_DAYS > 0 THEN

'     Проверка перехода на запись другого расчетного периода в текущем учетном.

'     Такая ситуация возможна при наличии перерасчетов отпуска задним числом.

      if VO_DAT_RP<>DATE_PACK(1,MONTH(CUR_RP()),YEAR(CUR_RP())) then exit do

'     Переключаем флаг контроля наличия отпуска по уходу за детьми в значение 1.

      PH_INT3=1

      PH_SRD_SM1=SUM_MIN_ZP( VO_DAT_RP )

      for i=DAY(VO_DAT_BEG_MON) to DAY(VO_DAT_END_MON)

        TBL[i]=1

      next i
'     Собираем значение для исходной суммы фонда (для отчета № 610)

      PH_SRD_SM2=PH_SRD_SM2+VO_FK_SM

'     Определяем процент начисления от минимальной зарплаты датировано

'     по датированному списку величин. Например, если 10% от 32% от

'     минимальной зарплаты, то достаточно ввести с 01/01/05 значение 0,032

'     (расчет: 10%*32%=0,1*0,32=0,032).

'     Настраивается Список датированных величин в Настройке-Справочники-

'     Списки величин.

'     PH_SRD_SM3=SDV_VALUE( Код списка, VO_DAT_RP ) - в примере код списка =1.

      PH_SRD_SM3=SDV_VALUE( 1, VO_DAT_RP )

     ENDIF

   loop while RL_NEXT() > 0

endif

if PH_INT3=1 then

  PH_INT1=0

  for i=1 to 31

'  Подсчитываем количество календарных дней в периоде, занятых отпуском

'  по уходу за ребенком

   PH_INT1=PH_INT1+TBL[i]

  next i

 ' MESSAGE("PROC= $PROC")

  FND_SUM = ROUND(PH_SRD_SM1 * PH_SRD_SM3 * PH_INT1/PH_INT2, 2)

 ' Исходная сумма для расчета (для отчета 610) - приравниваем сумме

' минимальной зарплаты на дату пропорционально календарным дням,

' на которые приходятся дни отпуска.

  FND_SUM_CLC=PH_SRD_SM1 * PH_INT1/PH_INT2

' Исходная сумма фонда (для отчета 610) - приравниваем сумме отпуска по

' уходу за ребенком до 3-х лет

  FND_SUM_SRC=PH_SRD_SM2

' Для формирования сумм фонда в отчете 245 по коду виду оплат

' "оплачиваемый отпуск по уходу за ребенком до 3-х лет" присвойте полю

' FND_CL_VO_CD значение, равное коду данного вида оплаты.

' FND_CL_VO_CD="код вида оплаты" - в примере код вида оплаты = 147

  FND_CL_VO_CD="147"

ELSE

  FND_SUM = 0

endif

